

Latin Square Design

Design of Experiments - Montgomery
Section 4-2

Latin Square Design

- Block on two nuisance factors
- One trt observation per block1
- One trt observation per block2
- Must have same number of blocks and treatments
- Two restrictions on randomization

$$y_{ijk} = \mu + \alpha_i + \tau_j + \beta_k + \epsilon_{ijk} \begin{cases} i = 1, 2, \dots, p \\ j = 1, 2, \dots, p \\ k = 1, 2, \dots, p \end{cases}$$

μ - grand mean

α_i - i th block 1 effect (row)

τ_j - j th treatment effect

β_k - k th block 2 effect (column)

$\epsilon_{ijk} \sim N(0, \sigma^2)$

- Completely additive model (no interaction)

Latin Square Design

- Design represented in $p \times p$ grid
- Randomization restrictions
 - One trt per row
 - One trt per column
- Shuffle rows and columns of standard square
- Examples

C	B	A
B	A	C
A	C	B

B	A	C
A	C	B
C	B	A

D	B	C	A
C	D	A	B
B	A	D	C
A	C	B	D

Proc PLAN

```

title 'Latin Square Design';
proc plan seed=12;
  factors rows=4 ordered cols=4 ordered / NOPRINT;
  treatments tmts=4 cyclic;
  output out=g
    rows cvals=('Day 1' 'Day 2' 'Day 3' 'Day 4') random
    cols cvals=('Lab 1' 'Lab 2' 'Lab 3' 'Lab 4') random
    tmts nvals=( 0 100 250 450 ) random;
proc tabulate;
  class rows cols;
  var tmts;
  table rows, cols*(tmts*f=6.) / rts=8;
run;
    
```

		cols			
		Lab 1	Lab 2	Lab 3	Lab 4
		tmts	tmts	tmts	tmts
		Sum	Sum	Sum	Sum
rows					
Day 1		450	100	0	250
Day 2		100	0	250	450
Day 3		0	250	450	100
Day 4		250	450	100	0

Partitioning the SS

- Rewrite observation as:

$$y_{ijk} = \bar{y}_{...} + (\bar{y}_{i..} - \bar{y}_{...}) + (\bar{y}_{.j.} - \bar{y}_{...}) + (\bar{y}_{..k} - \bar{y}_{...}) + (y_{ijk} - \bar{y}_{i..} - \bar{y}_{.j.} - \bar{y}_{..k} + 2\bar{y}_{...})$$

$$= \hat{\mu} + \hat{\alpha}_i + \hat{\tau}_j + \hat{\beta}_k + \hat{\epsilon}_{ijk}$$

- Partition SS_T into

$$p \sum (\bar{y}_{i..} - \bar{y}_{...})^2 + p \sum (\bar{y}_{.j.} - \bar{y}_{...})^2 + p \sum (\bar{y}_{..k} - \bar{y}_{...})^2 + \sum \sum \hat{\epsilon}_{ijk}^2$$

$$SS_{Row} + SS_{Treatment} + SS_{Col} + SS_E$$

- Under H_0 , all SS/σ^2 independent chi-squared RVs
- Usual F-test analysis
- Caution testing column and row effects

12-4

Analysis of Variance Table

Source of Variation	Sum of Squares	Degrees of Freedom	Mean Square	F_0
Rows	SS_{Row}	$p - 1$	MS_{Row}	
Treatment	$SS_{Treatment}$	$p - 1$	$MS_{Treatment}$	F_0
Column	SS_{Column}	$p - 1$	MS_{Column}	
Error	SS_E	$(p - 2)(p - 1)$	MS_E	
Total	SS_T	$p^2 - 1$		

$$SS_T = \sum \sum \sum y_{ijk}^2 - y_{...}^2 / p^2$$

$$SS_{Row} = \frac{1}{p} \sum y_{i..}^2 - y_{...}^2 / p^2$$

$$SS_{Treatment} = \frac{1}{p} \sum y_{.j.}^2 - y_{...}^2 / p^2$$

$$SS_{Column} = \frac{1}{p} \sum y_{..k}^2 - y_{...}^2 / p^2$$

$$SS_{Error} = \text{Use subtraction}$$

If $F_0 > F_{\alpha, p-1, (p-2)(p-1)}$ then reject H_0

12-5

Missing Values

- When missing
 - Design unbalanced
 - Orthogonality lost
 - Order of fit important
- Procedures
 - 1 Regression approach
 - Use Type III sum of squares
 - 2 Estimate missing value
 - Choose value to minimize SS_E
 - Take derivative and set equal to zero

$$y_{ijk} = \frac{p(y'_{i..} + y'_{.j.} + y'_{..k}) - 2y'_{...}}{(p-2)(p-1)}$$

12-6

Using SAS

Consider experiment to investigate the effect of 4 diets on milk production. There are 4 cows. Each lactation period the cows receive a different diet. Assume there is a washout period so previous diet does not affect future results. Will block on lactation period and cow.

```
options nocenter ls=75;options colors=(none);

data new;
  input cow period trt resp @@;
cards;
  1 1 1 38 1 2 2 32 1 3 3 35 1 4 4 33
  2 1 2 39 2 2 3 37 2 3 4 36 2 4 1 30
  3 1 3 45 3 2 4 38 3 3 1 37 3 4 2 35
  4 1 4 41 4 2 1 30 4 3 2 32 4 4 3 33
;

proc glm;
  class cow trt period;
  model resp=trt period cow;
  means trt / lines tukey;
  means period cow;
  output out=new1 r=res p=pred;

  symbol1 v=circle;
  proc gplot; plot res*pred;

proc univariate noprint;
  histogram res / normal (L=1 mu=0 sigma=est) kernel (L=2);
run;
```

12-7

Dependent Variable: resp

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	9	242.5625000	26.9513889	33.17	0.0002
Error	6	4.8750000	0.8125000		
Corrected Total	15	247.4375000			

Source	DF	Type I SS	Mean Square	F Value	Pr > F
trt	3	40.6875000	13.5625000	16.69	0.0026
period	3	147.1875000	49.0625000	60.38	<.0001
cow	3	54.6875000	18.2291667	22.44	0.0012



Source	DF	Type III SS	Mean Square	F Value	Pr > F
trt	3	40.6875000	13.5625000	16.69	0.0026
period	3	147.1875000	49.0625000	60.38	<.0001
cow	3	54.6875000	18.2291667	22.44	0.0012

Tukey's Studentized Range (HSD) Test for resp

Alpha	0.05
Error Degrees of Freedom	6
Error Mean Square	0.8125
Critical Value of Studentized Range	4.89559
Minimum Significant Difference	2.2064

	Mean	N	trt
A	37.5000	4	3
A	37.0000	4	4
B	34.5000	4	2
B	33.7500	4	1

12-8



12-9

Using Proc Mixed

- Sometimes, block should be considered random
 - Example: 4 cows randomly chosen from large herd
 - Want the inference to extend to the herd
 - Treat cow as a random blocking factor
- Sometimes measuring same EU over time/period
 - Example: Cow measured over 4 lactation periods
 - Are lactation periods closer together more similar?
 - Will treat as crossover design later in this topic
- Proc Mixed can incorporate both concepts into model

12-10

Random Effects

- Similar results as with RCBD
- Standard error for a mean: Proc GLM incorrect
- Standard error for a contrast: Proc GLM correct

```
proc glm;
class cow trt period;
model resp=cow trt period;
random cow;
lsmeans trt / stderr tdiff;
```

```
proc mixed;
class cow trt period;
model resp=trt period;
random cow;
lsmeans trt/ diff;
run;
```

12-11

The GLM Procedure

trt	resp LSMEAN	Standard Error	Pr > t	LSMEAN Number
1	33.7500000	0.4506939	<.0001	1
2	34.5000000	0.4506939	<.0001	2
3	37.5000000	0.4506939	<.0001	3
4	37.0000000	0.4506939	<.0001	4

i/j	Least Squares Means for Effect trt			
	1	2	3	4
1		-1.1767	-5.88348	-5.09902
		0.2839	0.0011	0.0022
2	1.176697		-4.70679	-3.92232
	0.2839		0.0033	0.0078
3	5.88348	4.706787		0.784465
	0.0011	0.0033		0.4626
4	5.09902	3.922323	-0.78446	
	0.0022	0.0078	0.4626	

The Mixed Procedure

Effect	trt	Least Squares Means				t Value	Pr > t
		Estimate	Std Error	DF			
trt	1	33.7500	1.1365	6	29.70	<.0001	
trt	2	34.5000	1.1365	6	30.36	<.0001	
trt	3	37.5000	1.1365	6	33.00	<.0001	
trt	4	37.0000	1.1365	6	32.56	<.0001	

Effect	trt	_trt	Differences of Least Squares Means				Pr > t
			Estimate	Std Error	DF	t Value	
trt	1	2	-0.7500	0.6374	6	-1.18	0.2839
trt	1	3	-3.7500	0.6374	6	-5.88	0.0011
trt	1	4	-3.2500	0.6374	6	-5.10	0.0022
trt	2	3	-3.0000	0.6374	6	-4.71	0.0033
trt	2	4	-2.5000	0.6374	6	-3.92	0.0078
trt	3	4	0.5000	0.6374	6	0.78	0.4626

Correlated Observations

- Residuals within an EU may be correlated
- Residuals closer in time may be more similar
- Can incorporate various correlation structures
- Represented as a $p \times p$ covariance matrix
- Main diagonal contains the variances
- Off-diagonal elements represent covariances
- Uncorrelated residuals (for $p = 4$ obs/cow)

$$\begin{bmatrix} \sigma^2 & 0 & 0 & 0 \\ 0 & \sigma^2 & 0 & 0 \\ 0 & 0 & \sigma^2 & 0 \\ 0 & 0 & 0 & \sigma^2 \end{bmatrix} = \sigma^2 \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

- 1st order autoregressive

$$\sigma^2 \begin{bmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{bmatrix}$$

SAS Code

```

/* Fit a simple correlation structure */
/* Similar to standard mixed analysis */

proc mixed covtest cl;
class cow trt period;
model resp=trt period / ddfm=kr outp=diag;
random cow;
repeated period / subject=cow type=simple;
lsmeans trt / diff;
run;

/* Fit an AR(1) correlation structure */

proc mixed covtest cl;
class cow trt period;
model resp=trt period / ddfm=kr outp=diag;
random cow;
repeated period / type=ar(1) subject=cow;
lsmeans trt / diff;
run;

```

Dimensions

Covariance Parameters	2
Columns in X	9
Columns in Z	4

Fit Statistics

-2 Res Log Likelihood	41.3
AIC (smaller is better)	45.3
AICC (smaller is better)	47.3
BIC (smaller is better)	44.1

Covariance Parameter Estimates

Parm	Subject	Estimate	Std Error	Z Value	Pr > Z	Alpha	Lower	Upper
cow		4.3542	3.7229	1.17	0.1211	0.05	1.346	73.99
period	cow	0.8125	0.4691	1.73	0.0416	0.05	0.337	3.94

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
trt	3	6	16.69	0.0026
period	3	6	60.38	<.0001

Least Squares Means

Effect	trt	Estimate	Std Error	DF	t Value	Pr > t
trt	1	33.7500	1.1365	3.82	29.70	<.0001
trt	2	34.5000	1.1365	3.82	30.36	<.0001
trt	3	37.5000	1.1365	3.82	33.00	<.0001
trt	4	37.0000	1.1365	3.82	32.56	<.0001

Differences of Least Squares Means

Effect	trt	_trt	Estimate	Std Error	DF	t Value	Pr > t
trt	1	2	-0.7500	0.6374	6	-1.18	0.2839
trt	1	3	-3.7500	0.6374	6	-5.88	0.0011
trt	1	4	-3.2500	0.6374	6	-5.10	0.0022
trt	2	3	-3.0000	0.6374	6	-4.71	0.0033
trt	2	4	-2.5000	0.6374	6	-3.92	0.0078
trt	3	4	0.5000	0.6374	6	0.78	0.4626

Dimensions	
Covariance Parameters	3
Columns in X	9
Columns in Z	4

Fit Statistics	
-2 Res Log Likelihood	41.2
AIC (smaller is better)	47.2
AICC (smaller is better)	52.0
BIC (smaller is better)	45.3

Covariance Parameter Estimates							
Cov Parm	Subject	Estimate	Std Error	Z Value	Pr > Z	Alpha	Lower Upper
cow		4.1459	3.7154	1.12	0.1322	0.05	1.2331 88.22
AR(1)	cow	0.2184	0.5794	0.38	0.7062	0.05	-0.9171 1.35
Residual		0.9292	0.7343	1.27	0.1029	0.05	0.3061 11.33

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
trt	3	5.04	8.65	0.0197
period	3	4.06	41.27	0.0017

Least Squares Means						
Effect	trt	Estimate	Std Error	DF	t Value	Pr > t
trt	1	33.7306	1.1424	3.81	29.53	<.0001
trt	2	34.4506	1.1424	3.81	30.16	<.0001
trt	3	37.5563	1.1424	3.81	32.87	<.0001
trt	4	37.0125	1.1424	3.81	32.40	<.0001

Effect	trt	_trt	Estimate	Std Error	DF	t Value	Pr > t
trt	1	2	-0.7200	0.7831	5.82	-0.92	0.3943
trt	1	3	-3.8257	0.8834	5.51	-4.33	0.0060
trt	1	4	-3.2819	0.7831	5.82	-4.19	0.0061
trt	2	3	-3.1057	0.7831	5.82	-3.97	0.0079
trt	2	4	-2.5619	0.8834	5.51	-2.90	0.0301
trt	3	4	0.5438	0.7831	5.82	0.69	0.5141

12-16

12-17

Replicated Latin Square

- Latin Square Design → df_E small
 - If 3 treatments → 2 df error
 - If 4 treatments → 6 df error
- Replication increases df_E without increasing trts
- Methods of replication
 - Same row and column blocks
 - New rows and same columns
 - Same rows and new columns
 - New rows and new columns
- Degrees of freedom depend on what is "new"/randomized
- Often include additional block - "replicate" effect

Replicate Square

- Same row/column blocks used in additional squares
- Usually includes replicate (e.g., time) effect

$$y_{ijkl} = \mu + \alpha_i + \tau_j + \beta_k + \delta_l + \epsilon_{ijkl} \quad \begin{cases} i = 1, 2, \dots, p \\ j = 1, 2, \dots, p \\ k = 1, 2, \dots, p \\ l = 1, 2, \dots, n \end{cases}$$

Replicated Rows (or columns)

- Different rows only
- Row effect often **nested** within square
- β_k can be different for each square

$$\sum \beta_{k(l)} = 0 \text{ instead of } \sum \beta_k = 0$$

$$y_{ijkl} = \mu + \alpha_{i(l)} + \tau_j + \beta_k + \delta_l + \epsilon_{ijkl} \quad \begin{cases} i = 1, 2, \dots, p \\ j = 1, 2, \dots, p \\ k = 1, 2, \dots, p \\ l = 1, 2, \dots, n \end{cases}$$

Source of Variation	Sum of Squares	Degrees of Freedom	Mean Square	F
Rows	SS_{Row}	$p - 1$		
Columns	SS_{Column}	$p - 1$		
Replicate	$SS_{Replicate}$	$n - 1$		
Treatment	$SS_{Treatment}$	$p - 1$	$MS_{Treatment}$	F_0
Error	SS_E	$(p - 1)(n(p + 1) - 3)$	MS_E	
Total	SS_T	$np^2 - 1$		

Source of Variation	Sum of Squares	Degrees of Freedom	Mean Square	F
Rows	SS_{Row}	$n(p - 1)$		
Columns	SS_{Column}	$p - 1$		
Replicate	$SS_{Replicate}$	$n - 1$		
Treatment	$SS_{Treatment}$	$p - 1$	$MS_{Treatment}$	F_0
Error	SS_E	$(p - 1)(np - 2)$	MS_E	
Total	SS_T	$np^2 - 1$		

12-18

12-19

Replicated Rows (or columns)

- Considered Latin Rectangle : “No replicate effect”
- np separate rows (n integer)

$$y_{ijk} = \mu + \alpha_i + \tau_j + \beta_k + \epsilon_{ijk} \quad \begin{cases} i = 1, 2, \dots, np \\ j = 1, 2, \dots, p \\ k = 1, 2, \dots, p \end{cases}$$

Source of Variation	Sum of Squares	Degrees of Freedom	Mean Square	F
Rows	SS_{Row}	$np - 1$		
Columns	SS_{Column}	$p - 1$		
Treatment	$SS_{\text{Treatment}}$	$p - 1$	$MS_{\text{Treatment}}$	F_0
Error	SS_E	$(p - 1)(np - 2)$	MS_E	
Total	SS_T	$np^2 - 1$		

12-20

Replicated Rows and Columns

- Have completely separate squares
- Usually row and column effect **nested** within square

$$y_{ijkl} = \mu + \alpha_{i(l)} + \tau_j + \beta_{k(l)} + \delta_l + \epsilon_{ijkl} \quad \begin{cases} i = 1, 2, \dots, p \\ j = 1, 2, \dots, p \\ k = 1, 2, \dots, p \\ l = 1, 2, \dots, n \end{cases}$$

Source of Variation	Sum of Squares	Degrees of Freedom	Mean Square	F
Rows	SS_{Row}	$n(p - 1)$		
Columns	SS_{Column}	$n(p - 1)$		
Replicate	$SS_{\text{Replicate}}$	$n - 1$		
Treatment	$SS_{\text{Treatment}}$	$p - 1$	$MS_{\text{Treatment}}$	F_0
Error	SS_E	$(p - 1)(n(p - 1) - 1)$	MS_E	
Total	SS_T	$np^2 - 1$		

12-21

Extensions

- Crossover Design
 - p treatments and p periods
 - np subjects (experimental units)
 - Analysis similar to replicated col Latin Square
 - Used in drug comparisons/physiology experiments
 - Delay between periods to remove residual effect
 - Model can handle residual effects ($p > 2$)
- Graeco-Latin Square Design (Section 4.3)
 - Superimpose two Latin Squares onto each other
 - Can block on three factors ($p \geq 3$ & $p \neq 6$)

12-22

Crossover Design

- Time is a blocking factor (usually called period)
- np subjects (S_k) receive p trts (τ_j) in p periods (P_i)
- Anticipate high level of variability between subjects
- Improve precision with several obs from each subj
- Commonly used for only 2,3, or 4 periods
- Should consider problem of subsequent use
- Analysis similar to Latin Rectangle/Replicated Cols

$$y_{ijk} = \mu + P_i + \tau_j + S_k + \epsilon_{ijk}$$

Consider 2 trt 2 period experiment with n subjects. Based on the model, the difference in trts for the two groups can be written

$$\begin{aligned} \text{Received Trt 1 first: } \text{diff}_{1k} &= (\tau_1 - \tau_2) + (P_1 - P_2) + (\epsilon_{11k} - \epsilon_{22k}) \\ \text{Received Trt 2 first: } \text{diff}_{2k} &= (\tau_2 - \tau_1) + (P_1 - P_2) + (\epsilon_{21k} - \epsilon_{12k}) \end{aligned}$$

Thus $\overline{\text{diff}}_1 - \overline{\text{diff}}_2$ estimates $2(\tau_1 - \tau_2)$ with variance $2\sigma^2/n$

12-23

Residual Effects

But what if there is a residual effect. In other words, the treatment effect is different for the different periods. The model can then be written

$$\begin{aligned} \text{Trt 1 first: } \text{diff}_{1k} &= (\tau_1 - \tau'_2) + (P_1 - P_2) + e_{1k} \\ \text{Trt 2 first: } \text{diff}_{2k} &= (\tau_2 - \tau'_1) + (P_1 - P_2) + e_{2k} \end{aligned}$$

Thus $\overline{\text{diff}}_1 - \overline{\text{diff}}_2$ estimates $(\tau_1 - \tau_2) + (\tau'_1 - \tau'_2)$. Cannot yield inference about both differences (i.e., confounded).

Can test for residual effect by looking at sums instead of differences. Subject variability incorporated into error ($\delta_{ijk} = \epsilon_{ijk} + S_k$)

$$\begin{aligned} \text{Trt 1 first: } \text{sum}_{1k} &= 2\mu + (\tau_1 + \tau'_2) + (P_1 + P_2) + \delta_{1k} \\ \text{Trt 2 first: } \text{sum}_{2k} &= 2\mu + (\tau_2 + \tau'_1) + (P_1 + P_2) + \delta_{2k} \end{aligned}$$

Thus $\overline{\text{sum}}_1 - \overline{\text{sum}}_2$ estimates $(\tau_1 - \tau_2) - (\tau'_1 - \tau'_2)$. Can check to see if different from zero. Problem if that occurs. Low power test because it incorporates subject to subject variability.

- More than 2 periods allows residual effect in model
- Not orthogonal, order of fit important (Type III)

$$y_{ijk} = \mu + P_i + \tau_j + S_k + r_{ij'} + \epsilon_{ijk} \quad \begin{cases} i = 1, 2, \dots, p \\ j = 1, 2, \dots, p \\ k = 1, 2, \dots, np \end{cases}$$

where $r_{ij'}$ only occurs when $i \neq 1$ and j' references the trt used in the previous period.

SAS Code

```
options nocenter ls=75;options colors=(none);

data new;
input cow period trt resp @@;
if period=1 then resid=0;
else resid=a;
resid1=0; resid2=0; resid3=0;
if resid=1 then resid1=1; if resid=4 then resid1=-1;
if resid=2 then resid2=1; if resid=4 then resid2=-1;
if resid=3 then resid3=1; if resid=4 then resid3=-1;
a=trt;
retain a;
cards;
1 1 1 38 1 2 2 32 1 3 3 35 1 4 4 33
2 1 2 39 2 2 3 37 2 3 4 36 2 4 1 30
3 1 3 45 3 2 4 38 3 3 1 37 3 4 2 35
4 1 4 41 4 2 1 30 4 3 2 32 4 4 3 33
;

proc print;

proc glm;
class cow period trt;
model resp=cow period trt resid1 resid2 resid3 /solution;
lsmeans trt / stderr pdiff cl;
run;
```

Obs	cow	period	trt	resp	resid	resid1	resid2	resid3
1	1	1	1	38	0	0	0	0
2	1	2	2	32	1	1	0	0
3	1	3	3	35	2	0	1	0
4	1	4	4	33	3	0	0	1
5	2	1	2	39	0	0	0	0
6	2	2	3	37	2	0	1	0
7	2	3	4	36	3	0	0	1
8	2	4	1	30	4	-1	-1	-1
.
16	4	4	3	33	2	0	1	0

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	12	244.6875000	20.3906250	22.24	0.0133
Error	3	2.7500000	0.9166667		
Corrected Total	15	247.4375000			

Source	DF	Type I SS	Mean Square	F Value	Pr > F
cow	3	54.6875000	18.2291667	19.89	0.0175
period	3	147.1875000	49.0625000	53.52	0.0042
trt	3	40.6875000	13.5625000	14.80	0.0265
resid1	1	0.5625000	0.5625000	0.61	0.4906
resid2	1	0.5208333	0.5208333	0.57	0.5057
resid3	1	1.0416667	1.0416667	1.14	0.3646

Source	DF	Type III SS	Mean Square	F Value	Pr > F
cow	3	46.0833333	15.3611111	16.76	0.0223
period	3	147.1875000	49.0625000	53.52	0.0042
trt	3	7.8409091	2.6136364	2.85	0.2062
resid1	1	0.3750000	0.3750000	0.41	0.5679
resid2	1	1.0416667	1.0416667	1.14	0.3646
resid3	1	1.0416667	1.0416667	1.14	0.3646

Source	DF	Type III SS	Mean Square	F Value	Pr > F
trt	3	7.8409091	2.6136364	2.85	0.2062
resid	3	2.1250000	0.7083333	0.77	0.5814

Parameter	Estimate	Standard Error	t Value	Pr > t
Intercept	33.00000000	0.95742711	34.47	<.0001
cow 1	0.62500000	0.82915620	0.75	0.5057
cow 2	2.00000000	0.82915620	2.41	0.0948
cow 3	5.37500000	0.82915620	6.48	0.0075
cow 4	0.00000000	.	.	.
period 1	8.00000000	0.67700320	11.82	0.0013
period 2	1.50000000	0.67700320	2.22	0.1135
period 3	2.25000000	0.67700320	3.32	0.0449
period 4	0.00000000	.	.	.
trt 1	-3.62500000	1.58771324	-2.28	0.1066
trt 2	-4.00000000	1.58771324	-2.52	0.0862
trt 3	-1.37500000	1.58771324	-0.87	0.4502
trt 4	0.00000000	.	.	.
resid1	0.75000000	1.17260394	0.64	0.5679
resid2	1.25000000	1.17260394	1.07	0.3646
resid3	-1.25000000	1.17260394	-1.07	0.3646

Residual Effect of A increases response 0.75 units
Residual Effect of B increases response 1.25 units
Residual Effect of C decreases response 1.25 units
Residual Effect of D decreases response 0.75 units

$$\text{LSMEAN trt 1} = 35.6875 + (-3.625 - .25(-3.625-4.000-1.375)) = 34.3125$$

= grand mean + trt effect

$$\text{LSMEAN trt 2} = 35.6875 + (-4.000 - .25(-3.625-4.000-1.375)) = 33.9375$$

= grand mean + trt effect

Designs Balanced For Residual Effects

- Consider the following two latin squares
- Suppose row=period and column=subject

D	C	B	A
C	D	A	B
B	A	D	C
A	B	C	D

D	C	B	A
C	A	D	B
B	D	A	C
A	B	C	D

- (Left) C → D twice, A → D once, B → D never
- (Right) Each trt follows each other trt once
- Right square **balanced** for residual effects
- If p even, can be balanced using p subjects
- If p odd, need multiple of $2p$ subjects

Least Squares Means				
trt	resp LSMEAN	Standard Error	Pr > t	LSMEAN Number
1	34.3125000	1.0013012	<.0001	1
2	33.9375000	1.0013012	<.0001	2
3	36.5625000	1.0013012	<.0001	3
4	37.9375000	1.0013012	<.0001	4

Least Squares Means for effect trt
Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: resp				
i/j	1	2	3	4
1		0.8285	0.2514	0.1066
2	0.8285		0.1968	0.0862
3	0.2514	0.1968		0.4502
4	0.1066	0.0862	0.4502	

Least Squares Means for Effect trt

Difference Between Means				
i	j	95% Confidence Limits for LSMean(i)-LSMean(j)		
1	2	0.375000	-4.677812	5.427812
1	3	-2.250000	-7.302812	2.802812
1	4	-3.625000	-8.677812	1.427812
2	3	-2.625000	-7.677812	2.427812
2	4	-4.000000	-9.052812	1.052812
3	4	-1.375000	-6.427812	3.677812